
Om LAX i Datorteknik

Den laborativa examinationen (LAX) är en avslutande examinationspunkt i labserien. LAX genom-
förs enskilt och utan samarbete med andra studenter.

Vid LAX-tillfället är datorerna i labsalen i examinationsläge vilket innebär att nät̊atkomst är av-
stängd. Inloggning sker lokalt p̊a datorn med eget användarnamn och lösenord.

Hjälpmedel P̊a LAX-en behöver du ta med dig penna. Kladdpapper finns i labsalen. P̊a varje
labplats finns dessutom en kopia av kursens Infoblad som delades ut p̊a första föreläsningen. Linjal
har du p̊a LiU-kortet. Inget övrigt skall behövas.

Hårdvara Ingen h̊ardvara utöver den som använts i laborationsserien kommer användas. Tan-
gentbordet och knappar är vanliga insignaler och lysdioder och displayer är utenheter. Högtalare
kommer inte användas (tack och lov), inte heller oscilloskopet eller IR-mottagaren.

Alla laxar inneh̊aller åtminstone följande moment:

• konfigurera stack

• sätta in- och utportar

• villkorliga hopp

• n̊agon databehandling

• h̊ardvara fr̊an labbarna

Exempellaxar finns p̊a hemsidan.

Ingen LAX kräver avbrott för sin lösning men en del kan bli enklare om avbrott används.

Det finns en uppsättning LAX-uppgifter och de slumpas till varje tillfälle.

Ett LAX-tillfälle är 90 minuter l̊angt och g̊angen vid laxtillfället är denna:

1. Bladet med LAX delas ut. Labtiden börjar.

2. När du har en lösning, p̊akalla assistentens uppmärksamhet.

3. Assistenten testar funktionen enligt LAX-bladet samt kontrollerar i koden att h̊ardvaruinitie-
ringen utförts i en subrutin:

a) Vid korrekt funktion och kod: Du antecknas med G i listan, är färdig, monterar ner
uppkopplingen och lämnar lokalen.

b) Vid felaktig funktion och/eller kod är du inte klar och f̊ar fortsätta med uppgiften. Se
punkt 2 ovan.

4. Vid full tid avlsutar assistenten laxtillfället.

Inget facit till LAX-en kommer delas ut. Inte heller kommer lösningar att diskuteras varken under
själva LAX-tillfället eller senare.

Lämna kvar LAX-bladet och Infobladet p̊a labplatsen efter LAX-en. Eventuella anteckningar skall
lämnas i papperskorgen innan du lämnar salen.



Detta dokument inneh̊aller förslag p̊a n̊agra representativa demonstrationslaxar. Uppgifterna är
konstruerade s̊a att de ska inneh̊alla sekvens, iteration och selektion enligt JSP. Dessutom bör de
komma ih̊ag n̊agon tidigare händelse. De ”skarpa” LAX-arna är av samma komplexitetsgrad, har
samma eller liknande h̊ardvara men är inte dessa uppgifter. Laxarna kräver inte avbrott men f̊ar
användas.

Inlärningen sker i den kreativa processen i hjärnan när du själv konstruerar en lösning. Titta inte p̊a
lösningsförslagen om du inte har ett eget förslag att jämföra med!

Namn In-enhet Ut-enhet

LAX-DEMO 1 Hextangentbord 2 x 7-segmentsdisplay
LAX-DEMO 2 2 x Tryckknapp 2 x 7-segmentsdisplay
LAX-DEMO 4 Hextangentbord 2 x 7-segmentsdisplay
LAX-DEMO 5 Hextangentbord Lysdioddisplay

Öva s̊a att du kan ta fram en fungerande lösning p̊a 30–45 minuter.



Datorteknik LAX
Förslag p̊a lösningsg̊ang

Först och främst är det avgörande att ha en plan! Lös inte hela uppgiften p̊a en g̊ang, det är
alltid fel approach, i all programmering. Man m̊aste g̊a i sm̊a steg. Till exempel dessa steg (i
en hypotetisk LAX med ett hextangentbord och lysdiodsrampen med 8 dioder där uppgiften är
att skicka tangentbordsvärdet till dioderna ”rakt av” eller inverterat beroende p̊a en ytterligare
insignal fr̊an en tryckknapp):

1. Rita skiss/schema över uppkopplingen med alla väsentliga kopplingar!

2. Koppla in lysdioderna. Kanske PortA bit 7 – 0.

3. Konfigurera in- och utportar.

ldi r16 ,$FF
out DDRA ,r16
ldi r16 ,$00
out DDRB ,r16

4. Kontrollera att portarna är korrekt konfigurerade genom att tända och släcka dioderna.
Kod för att testa dioderna kan vara n̊agot s̊a enkelt som

ldi r16 ,$AA
out PORTA ,r16
ldi r16 ,$55
out PORTA ,r16

som du kompilerar, och sen stegar igenom för att tända och släcka alla dioder. Nu vet
du att hela kedjan fungerar (kompilering/nedladdning/stegning/portar/lysdioder). Hoppa
inte över denna testning.

5. Koppla in tangentbordets fem bitar, STROBE samt D3–D0 (typiskt p̊a pinnarna 4–0). Kon-
trollera att dalia-kortets lysdioder tänds som du förväntar dig när du trycker p̊a olika
tangenter dvs utan programmet inblandat. OK, tangentbordet funkar.

6. Kontrollera tangentbordets funktion genom att skriva programkod för att testa. För att
veta om tangentbordet lästs in korrekt skickar du lämpligen ut inlästa värdet p̊a dioderna:

TEST:
in r16 ,PINB
out PORTA ,r16
jmp TEST

Kompilera/stega osv. OK, du kan läsa in tangentbordets bitar.

Om vi l̊anar begrepp fr̊an reglertekniken har du nu ett system som är styrbart (du vet
att du kan läsa in värden) och observerbart (du vet att du kan skicka ut värden). Utan
b̊ada dessa kommer systemet aldrig att fungera.

7. Koppla in tryckknappen. Skriv ett program som läser av den biten (typiskt p̊a pinne 7 p̊a
porten du redan använder som ing̊ang, men det är inte slagit i sten) och gör n̊agot med
displayen bara för att kontrollera att du har styrbarhet fr̊an tryckknappen:

KNAPP_KONTROLL:
ldi r16 ,$FF
sbis PINB ,7
clr r16
out PORTA ,r16
jmp KNAPP_KONTROLL



Kör du programmet i full fart ska alla dioder tändas vid nedtryckning av knappen och
släckas annars.

8. Sista steget här är att inte tända/släcka alla dioder utan bara de som tangentbordet vill
enligt uppgiften ovan. Bäst är nu om hela programmet redan är uppdelat i klippfärdiga
bitar. Här menar jag först̊as subrutiner. Skriv subrutiner!
”Bra” rutiner som du kanske redan har vid det här laget är:

• READ_KBD (läser tangentbord och returnerar inläst värde i r16)

• EMIT (skicka ut värdet i r16 till dioderna)

Programmet kan d̊a se ut s̊a här:

MAIN_LOOP:
call READ_KBD
call EMIT
jmp MAIN_LOOP

Hela koden kan skrivas som (med tv̊a rader för att känna av tryckknappen):

MAIN_LOOP:
call READ_KBD
sbis PINB ,7
clr r16
call EMIT
jmp MAIN_LOOP

Provkör. Asch, funkar bara nästan! ”clr r16” gör ju att displayen släcks om tangenten
inte är nedtryckt. Prova med complement-instruktionen istället:

MAIN_LOOP:
call READ_KBD
sbis PINB ,7
com r16 ; complement istf clr
call EMIT
jmp MAIN_LOOP

Provkör. Asch, funkar bara nästan! Inverterat värde skulle ut när man tryckte p̊a knappen,
inte när man inte tryckte p̊a knappen. En sista ändring d̊a:

MAIN_LOOP:
call READ_KBD
sbic PINB ,7 ; skip on no key
com r16
call EMIT
jmp MAIN_LOOP

Det där borde funka.

Det blev l̊angt det här men avsikten är att p̊aminna att man m̊aste göra allt i sm̊a steg. Varje steg
är i sig inte s̊a komplicerat, men att sätta ihop till subrutiner, ändra clr→com och sbis→sbic
kräver m̊anga bollar i luften och en koncentration och lugn som kanske inte infinner sig vid
examinationen.

För att fullfölja subrutinstänket skulle programmet slutligen kunna vara

MAIN_LOOP:
call READ_KBD
call INVERT_ON_KEY
call EMIT
jmp MAIN_LOOP

där INVERT_ON_KEY utför inverteringen som önskat p̊a argumentet r16 och returnerar rätt värde
i r16 à la ”r16 = INVERT_ON_KEY(r16)”.

En minimal lösning (utan subrutiner, s̊a inte giltig p̊a LAX) kan göras p̊a mindre än 20 bytes.
Kan du hitta den?

-o-O-o-



Datorteknik
LAX-DEMO 1

Tidsomfattning: 90 minuter
inkl redovisning

Uppgift I labsatsen finns ett hexadecimalt tan-
gentbord. Det ger ut fyra bitar data (d,c,b,a) vid
nedtryckt tangent men även en strobe-signal som
är hög s̊a länge n̊agon tangent är nedtryckt.

Din uppgift är att läsa av det hexadecimala tan-
gentbordet, en siffra åt g̊angen, och presentera
dess decimala motsvarighet p̊a tv̊a sjusegments-
displayer, tiotalssiffran till vänster och entalssiff-
ran till höger. Displayerna kan visa de hexadeci-
mala siffrorna 0-F, men ska här bara visa 0-9.
Senaste decimaltal ska kontinuerligt visas tills en
ny siffra trycks ned p̊a tangentbordet.

Tangentbordet kan inte ge flera utsignaler även
om flera tangenter trycks ned samtidigt, följakt-
ligen behöver inte programmet ta n̊agon hänsyn
till detta fall.

Obs! H̊ardvaruinitieringen m̊aste utföras som en
subrutin.

Hårdvara

• labsats

• sjusegmentsdisplay

• hexadecimalt tangentbord

Funktionskontroll och examination Funktio-
nen prioriteras! N̊agon kodgranskning, utöver kon-
troll av att h̊ardvaruinitieringen är utförd som
subrutin, kommer inte ske. Funktionen kontrol-
leras genom upprepade tryckningar p̊a det hexa-
decimala tangentbordet och kontroll p̊a sjuseg-
mentsdisplayen. Nöjaktig funktion resulterar i god-
känd LAX.

Sjusegmentsdisplayen Matningsspänning är
U (5 V) och 0 (0 V). Indata till respektive seg-
ment är de fyra bitarna d, c, b, a.

Tangentbordet med kopplingsplatta Matnings-
spänning +5 V p̊aförs pinne 1, 0 V pinne 8. Ut-
data, fyra bitar, återfinns p̊a pinnarna 3, 5, 7 och
9. Strobe-signalen är pinne 2. Stroben är hög s̊a
länge n̊agon knapp är nedtryckt. Datat kommer
samtidigt och ligger kvar tills nästa knappned-
tryckning:

DATA DATA DATADATA
Tryck Tryck Tryck



Datorteknik
LAX-DEMO 2

Tidsomfattning: 90 minuter
inkl redovisning

Uppgift I labsatsen finns tv̊a tryckknappar.
Dessa ger en positiv och en negativ flank som
utsignal fr̊an var sina stift, för respektive tryck-
knapp.

Din uppgift är att räkna antalet nedtryckning-
ar av den vänstra tryckknappen. När den hög-
ra tryckknappen trycks ned ska detta antal visas
p̊a en sjusegmentsdisplay, och fortsätta att visas
även efter att den högra tryckknappen släppts
upp. Därefter ska man kunna börja om med att
räkna nedtryckningar av den vänstra tryckknap-
pen. Displayerna kan visa de hexadecimala siff-
rorna 0-F. Trycker man mer än 15 g̊anger p̊a den
vänstra tryckknappen s̊a ska displayen visa F, dvs
1510 hexadecimalt.

Obs! H̊ardvaruinitieringen m̊aste utföras som en
subrutin.

Hårdvara

• sjusegmentsdisplay

• tryckknappar

Funktionskontroll och examination Funktio-
nen prioriteras! N̊agon kodgranskning, utöver kon-
troll av att h̊ardvaruinitieringen är utförd som
subrutin, kommer inte ske. Funktionen kontrolle-
ras genom upprepade tryckningar p̊a tryckknap-
parna och kontroll p̊a sjusegmentsdisplayen. Nöj-
aktig funktion resulterar i godkänd LAX.

Sjusegmentsdisplayen Matningsspänning är
U (5 V) och 0 (0 V). Indata till respektive seg-
ment är de fyra bitarna d, c, b, a.

Tryckknappar Matningsspänning är U (5 V)
och 0 (0 V). Varje knapp ger en positiv och en
negativ flank som utsignal fr̊an var sina stift d̊a
knappen trycks ned. Utsignalen återg̊ar sedan när
knappen släpps upp.



Datorteknik
LAX-DEMO 4

Tidsomfattning: 90 minuter
inkl redovisning

Uppgift: I labsatsen finns ett hexadecimalt
tangentbord. Det ger ut fyra bitar data (d,c,b,a)
vid nedtryckt tangent men även en strobe-signal
som är hög s̊a länge n̊agon tangent är nedtryckt.

Din uppgift är att visa nedtryckta decimala tan-
gentvärden fr̊an tangentbordet p̊a vänster alter-
nativt höger indikator p̊a sjusegmentsdisplayen.
Med tangenten F ska man kunna växla (toggla)
indikator s̊a att efterföljande tangentvärden ham-
nar till vänster ifall höger indikator tidigare an-
vändes, och vice versa. Gamla värden ska dock
alltid ligga kvar tills dom ersätts av nya fr̊an tan-
gentbordet. Tangenterna A, B, C, D och E ska
inte ha n̊agon funktion eller p̊averkan.

Obs! H̊ardvaruinitieringen m̊aste utföras som en
subrutin.

Hårdvara

• labsats

• sjusegmentsdisplay

• hexadecimalt tangentbord

Funktionskontroll och examination Funktio-
nen prioriteras! N̊agon kodgranskning, utöver kon-
troll av att h̊ardvaruinitieringen är utförd som
subrutin, kommer inte ske. Funktionen kontrol-
leras genom upprepade tryckningar p̊a det hexa-
decimala tangentbordet och kontroll p̊a sjuseg-
mentsdisplayen. Nöjaktig funktion resulterar i god-
känd LAX.

Sjusegmentsdisplay Displayen har tv̊a sju-
segments indikatorer. Matningsspänning är U (5 V)
och 0 (0 V). Indata till respektive segment är de
fyra bitarna d, c, b, a.

Tangentbordet med kopplingsplatta Matnings-
spänning +5 V p̊aförs pinne 1, 0 V pinne 8. Ut-
data, fyra bitar, återfinns p̊a pinnarna 3, 5, 7 och
9. Strobe-signalen är pinne 2. Stroben är hög s̊a
länge n̊agon knapp är nedtryckt. Datat kommer
samtidigt och ligger kvar tills nästa knappned-
tryckning:

DATA DATA DATADATA
Tryck Tryck Tryck



Datorteknik
LAX-DEMO 5

Tidsomfattning: 90 minuter
inkl redovisning

Uppgift: I labsatsen finns ett hexadecimalt
tangentbord. Det ger ut fyra bitar data (d,c,b,a)
vid nedtryckt tangent men även en strobe-signal
som är hög s̊a länge n̊agon tangent är nedtryckt.

Din uppgift är att visa ett nedtryckt tangentvär-
de binärt p̊a lysdioddisplayens fyra mest vänst-
ra lysdioder. Samma tangentvärde ska även vi-
sas p̊a dom fyra mest högra lysdioderna, inverte-
rat (obs ej speglat) eller normalt. Inverterad eller
normal visning för dom fyra mest högra lysdio-
derna togglas med tangent 0, som ocks̊a ska visas
p̊a samma sätt som övriga tangentvärden. Dvs,
om normal visning r̊ader ska lysdioderna 7–4 och
lysdioderna 3–0 visa samma sak, men om inver-
terad visning r̊ader ska lysdioderna 3–0 visa det
inverterade värdet av lysdioderna 7–4, vilka allts̊a
alltid ska visa tangentvärdet.

Obs! H̊ardvaruinitieringen m̊aste utföras som en
subrutin.

Hårdvara

• labsats

• lysdioddisplay

• hexadecimalt tangentbord

Funktionskontroll och examination Funktio-
nen prioriteras! N̊agon kodgranskning, utöver kon-
troll av att h̊ardvaruinitieringen är utförd som
subrutin, kommer inte ske. Funktionen kontrol-
leras genom upprepade tryckningar p̊a det hexa-
decimala tangentbordet och kontroll p̊a lysdiod-
displayen. Nöjaktig funktion resulterar i godkänd
LAX.

Lysdioddisplay Matningsspänning är U (5 V)
och 0 (0 V). Lysdioderna styrs individuellt av in-
g̊angarna 7 till 0 nederst p̊a plattan.

Tangentbordet med kopplingsplatta Matnings-
spänning +5 V p̊aförs pinne 1, 0 V pinne 8. Ut-
data, fyra bitar, återfinns p̊a pinnarna 3, 5, 7 och
9. Strobe-signalen är pinne 2. Stroben är hög s̊a
länge n̊agon knapp är nedtryckt. Datat kommer
samtidigt och ligger kvar tills nästa knappned-
tryckning:

DATA DATA DATADATA
Tryck Tryck Tryck





Lösningsförslag

Instruktioner

Inlärningen sker i den kreativa processen i hjärnan när du själv konstruerar en lösning. Titta inte p̊a
förslagen nedan om du inte har ett eget förslag att jämföra med!

Man lär sig koda bättre genom att läsa mycket kod. Jämför förslagen med din egen lösning och
förbättra dem b̊ada. Vad kan göras för att f̊a mer lättläst kod? Är strukturen den bästa? Skulle
koden tjäna p̊a globala konstanter? Variabelnamn? Finns det alternativa lösningsmetoder?

Vid LAX-tillfället sker visserligen ingen kodgranskning men man tjänar änd̊a p̊a att ha ett struktu-
rerat angreppssätt med subrutiner och bra namngivning av labels.

Namn In-enhet Ut-enhet

LaxDemo1.asm Hextangentbord 2 x 7-segmentsdisplay
LaxDemo2.asm 2 x Tryckknapp 2 x 7-segmentsdisplay
LaxDemo2a.asm Dalia Dalia
LaxDemo4.asm Hextangentbord 2 x 7-segmentsdisplay
LaxDemo5.asm Hextangentbord Lysdioddisplay
LaxDemo5mini.asm Hextangentbord Lysdioddisplay

LabDemo5mini.asm är ett försök att konstruera en resurssn̊al lösning. Här har man eliminerat kod
genom att bland annat koppla om h̊ardvaran och ta bort — i det här fallet — onödiga instruktioner.
Lösningen tillhör kategorin ”ful kod”.



/* LaxDemo1.asm
* Compiles to 50 bytes (42 if call etc)
*/

COLD:
ldi r16 ,HIGH(RAMEND)
out SPH ,r16
ldi r16 ,LOW(RAMEND)
out SPL ,r16
call HW_INIT

MAIN:
sbis PINA ,4 ; wait for strobe/key press
jmp MAIN
in r16 ,PINA ; read key
andi r16 ,$0F
cpi r16 ,10
brmi PRINT
subi r16 ,$FA

PRINT:
out PORTB ,r16

WAIT:
sbic PINA ,4 ; wait for key release
jmp WAIT
jmp MAIN ; process next digit

; --- Config I/O
HW_INIT:

ldi r16 ,0
out DDRA ,r16 ; PORTA <4> strobe , PORTA <3-0> data
dec r16
out DDRB ,r16 ; PORTB <7-0> display
ret



/* LaxDemo2.asm
* Compiles to 70 bytes (62 if call etc)
*/

COLD:
ldi r16 ,HIGH(RAMEND)
out SPH ,r16
ldi r16 ,LOW(RAMEND)
out SPL ,r16
call HW_INIT

WARM:
call GETKEYS
sbrc r16 ,1
inc r17 ; left pressed
sbrc r16 ,0
call SHOWIT ; right pressed
jmp WARM

SHOWIT:
cpi r17 ,$0F
brmi SHOWIT2
ldi r17 ,$0F

SHOWIT2:
out PORTB ,r17
; clr r17 ; show progress
ret

GETKEYS:
in r16 ,PIND
andi r16 ,$03
cpi r16 ,$00
brne GETKEYS ; wait for release

GETKEYS2:
in r16 ,PIND
andi r16 ,$03
cpi r16 ,$00
breq GETKEYS2 ; wait for press
ret

HW_INIT:
ldi r16 ,$00
out DDRD ,r16 ; bit1 - left , bit0 - right
ldi r16 ,$FF
out DDRB ,r16
clr r17 ; no sum yet
ret



/* LaxDemo2a.asm
* Compiles to 72 bytes (64 if call etc)
* Version for Dalia
* Input: Buttons INT1 and INT0
* Output: on -board LED
*/

COLD:
ldi r16 ,HIGH(RAMEND)
out SPH ,r16
ldi r16 ,LOW(RAMEND)
out SPL ,r16
call HW_INIT

WARM:
call GETKEYS
sbrs r16 ,3
inc r17 ; left pressed
sbrs r16 ,2
call SHOWIT ; right pressed
jmp WARM

SHOWIT:
cpi r17 ,$0F
brmi SHOWIT2
ldi r17 ,$0F

SHOWIT2:
out PORTB ,r17
; clr r17 ; show progress
ret

GETKEYS:
in r16 ,PIND
andi r16 ,$0C
cpi r16 ,$0C
brne GETKEYS ; wait for release

GETKEYS2:
in r16 ,PIND
andi r16 ,$0C
cpi r16 ,$0C
breq GETKEYS2 ; wait for press
ret

HW_INIT:
ldi r16 ,$00
out DDRD ,r16
ldi r16 ,$FF
out DDRB ,r16
out PORTD ,r16 ; pull -up, PD3 left , PD2 right
clr r17 ; no sum yet
ret



/* LaxDemo4.asm
* Compiles to 86 bytes (72 if rcall etc)
*/

COLD:
ldi r16 ,HIGH(RAMEND)
out SPH ,r16
ldi r16 ,LOW(RAMEND)
out SPL ,r16
call HW_INIT

WARM:
call GETKEY
cpi r16 ,$0F ; "F"?
brne NO_TOGGLE
com r18 ; yep!

NO_TOGGLE:
cpi r16 ,10 ; 0-9?
brpl WARM ; A-F no update
in r17 ,PORTB ; get displayed
sbrs r18 ,0 ; 0 -> right , 1 -> left
jmp RIGHT

LEFT:
andi r17 ,$0F ; clear left
swap r16 ; put digit in place
jmp SHOWIT

RIGHT:
andi r17 ,$F0 ; clear right

SHOWIT:
or r17 ,r16 ; merge
out PORTB ,r17 ; and display
jmp WARM

; --- GETKEY returns key in r16
GETKEY:

sbic PINA ,4 ; wait for release
jmp GETKEY

GETKEY2:
sbis PINA ,4
jmp GETKEY2 ; wait for press
in r16 ,PINA ; get key
andi r16 ,$0F
ret

; --- I/O
; PA4 STROBE
; PA3 -0 Data
; PB7 -4 Left digit
; PB3 -0 Right digit

HW_INIT:
ldi r16 ,$00
out DDRA ,r16
ldi r16 ,$FF
out DDRB ,r16
ldi r16 ,$00
out PORTB ,r16 ; "00"
clr r18 ; toggle byte
ret



/* LaxDemo5.asm
* Compiles to 70 bytes (60 if rcall etc)
*/

COLD:
ldi r16 ,HIGH(RAMEND)
out SPH ,r16
ldi r16 ,LOW(RAMEND)
out SPL ,r16
rcall HW_INIT

WARM:
rcall GETKEY
mov r17 ,r16 ; r16 rightmost
brne NO_ZERO
com r19 ; was a "0"

NO_ZERO:
cpi r19 ,0
breq WARM3
ldi r18 ,$0F ; invert right
eor r16 ,r18

WARM3:
swap r17
or r16 ,r17
out PORTB ,r16
rjmp WARM

; --- GETKEY Return pressed key in r16
GETKEY:

sbic PINA ,4 ; wait for release
rjmp GETKEY

GETKEY2:
sbis PINA ,4 ; wait for press
rjmp GETKEY2
in r16 ,PINA
andi r16 ,$0F ; return key
ret

; --- I/O init , initial state
HW_INIT:

clr r16
out DDRA ,r16
ldi r16 ,$FF
out DDRB ,r16
clr r19 ; 0 -> normal , 1 -> inverted
ret



/*
* LaxDemo5mini.asm
* Attempt at minimal , and hence ugly , code
* Compiles to 32 bytes
*/

COLD:
ldi r16 ,HIGH(RAMEND)
out SPH ,r16
ldi r16 ,$FF
out DDRB ,r16

WARM:
sbic PIND ,0 ; wait for release
rjmp WARM

KEY:
sbis PIND ,0 ; wait for press
rjmp KEY
in r16 ,PINA
cpi r16 ,0
brne CONT
com r17 ; was a "0"

CONT:
sbrc r17 ,0
eor r16 ,r17
out PORTB ,r16
rjmp WARM


